

    

        pattern_metonyms

        v0.8.0



    


  

    Table of contents

    
      



      	Overview


      	Changelog





        	
          Modules
          


      	PatternMetonyms





        



      

    

  

    PatternMetonyms

Attempt at implementing Pattern Synonyms from https://www.microsoft.com/en-us/research/wp-content/uploads/2016/08/pattern-synonyms-Haskell16.pdf but obviously missing a ton of check, most notably from the type checker.
Consider this as a personal project project until 1.0.0, you may open issues, but pull requests will be closed without consideration.
Warning
This library hides a lot of code, please keep yourself aware that by using it, you are trading a lot of runtime efficiency for a lot of (unmeasured) expressivness power.
Installation
The package can be installed by adding pattern_metonyms to your list of dependencies in mix.exs:
def deps do
  [
    {:pattern_metonyms, "~> 0.8.0"}
  ]
end
The docs can be found at https://hexdocs.pm/pattern_metonyms.
Examples
It can be hard to see at first glance why this would be interesting, so I'm going to try to fill example usages over time.
Stacks
Stacks are easy to implement, and easy to manipulate: just use a list !
You can use it raw:
	when you want to push something, just cons an element: new_stack = [x | stack]

	when you want to pop something, just unconse an element: [x | new_stack] = stack


We can improve the code by using functions, in order to give names to the operations.
This is useful because it gives us a vocabulary, thus when we encounter the code,
we don't have to refer to the context, or hold it in our head in order to follow the code:
	when you want to push something, just call the appropriate function: new_stack = Stack.push(stack, x)
	when you want to pop something, call the appropriate function and match on the result: {x, new_stack} = Stack.pop(stack)
This also has the advantage of functions being first class values, you can pass them around.

But has seen, we have to match while knowing what the result of the function will look like.
This is fine but it still presents some overhead, both in the execution compared to the raw version, and in the expressivity because of the necessity to match on the tuple.  
This slight expressivity impediment become more apparent when we want to use pattern matching to control the flow of execution (unless you define more functions to expand the vocabulary of what's possible).
Suppose you want to evaluate different expressions depending on if the stack is empty or not:
	when raw:{x, new_stack} = case stack do
  [] -> {default_value, stack}
  [x | new_stack] -> {x, new_stack}
end

	with functions:{x, new_stack} = case Stack.pop(stack) do
  :empty -> {default_value, stack}
  t -> t
end


This is one possibility, but what if Stack.pop/1 was defined more straightforwardly ? You would have to check whether the stack is empty before using it:
{x, new_stack} = if Stack.empty?(stack) do {default_value, stack} else Stack.pop(stack) end
Or you could have another version that directly handle the default value, with the potential of introducing some ambiguity if it was defined with a default argument itself.
So yes, this is a toy example, so it is absolutly possible to easily define a nice API with functions,
but it is possible with macros to define some vocabulary that can mix well with the pattern matching constructs.
We this library, you can define a vocabulary, that behaves as the raw code:
defmodule Stack do
  import PatternMetonyms

  pattern empty = []
  pattern push(stack, x) = [x | stack]
  pattern pop(x, stack) = [x | stack]
end
Assuming the module Stack was require'd:
	when you want to push something: new_stack = Stack.push(stack, x)
	when you want to pop something: Stack.pop(x, new_stack) = stack
	when you want to do some control flow:{x, new_stack} = case stack do
  Stack.empty() -> {default_value, stack}
  Stack.pop(x, stack) -> {x, stack}
end


That's all fine and dandy, but you don't need to use this library to define these kinds of macro. (Though it isn't as nice looking. :P)
So, on to the next part !
Queues
Queues aren't as simple as stacks, we don't have a ready made data structure for them we can pattern match on.
But we have the erlang module :queue that allows us to construct and manipulate erlang primitives in order to have something that acts like a queue.
Like the function section of the stacks, we have a vocabulary available to us to ease the understanding of the code.
In the same way, we can't directly use it with pattern matching. But it's not possible to define macros that mix well with the pattern matching construct.
Using this library, you can use new control flow constructs, the more interesting one being view/2. The others building on top of it.
Note:
  Unfortunatly it's currently too hard for me to define an equivalent to the match operator named =/2. (Mainly because I haven't made the effort.)
  Because of that, it's difficult to do something as nice as raw code for non conditional execution.
Using view patterns, you can directly pattern match with the functions of the module :queue:
	for control flow:{x, new_queue} = view queue do
  (:queue.is_empty() -> true) -> {default_value, queue}
  (:queue.out() -> {{:value, x}, queue}) -> {x, queue}
end
Note that :queue.out/1 returns {:empty, queue} in case the queue is empty, allowing to directly pattern match on :queue.out/1's result:{x, new_queue} = case :queue.out(queue) do
  {:empty, queue} -> {default_value, queue}
  {{:value, x}, queue} -> {x, queue}
end


Using patterns definition with view:
defmodule Queue do
  import PatternMetonyms

  pattern empty <- (:queue.is_empty() -> true)
  pattern pop(x, queue) <- (:queue.out() -> {{:value, x}, queue})
end
	for control flow, assuming Queue was require'd:{x, new_queue} = view queue do
  Queue.empty() -> {default_value, queue}
  Queue.pop(x, queue) -> {x, queue}
end


This become even more interesting with recursion (though this is still a toy example).
Say :queue.to_list/1 doesn't exist, neither :queue.fold/3, how would you go about defining it ?
	raw:defmodule Queue do
  # [...]

  def to_list(queue) do
    case :queue.out(queue) do
      {:empty, _} -> []
      {{:value, x}, queue} -> [x | to_list(queue)
    end
  end
end

	with patterns and defv/1:defmodule Queue do
  # [...]

  defv to_list(empty()), do: []
  defv to_list(pop(x, queue)), do: [x | to_list(queue)]
end


I hope you are now starting to get an idea about how to play with this lib.
More to come when the inspiration does.


  

    Changelog

All notable changes to this project will be documented in this file. (Hopefully I won't forget much.)
Unreleased
Changed
	Error message improvements when patterns are used improperly.

0.8.0 - 2022-06-21
Added
	Added PatternMetonyms.matchv?/2 as an equivalent to match?/2.

Changed
	Doc improvements.

0.7.0 - 2022-06-20
Added
	Added PatternMetonyms.fit/2 as an equivalent to =/2.
	Added PatternMetonyms.fit/1 to convert =/2 to PatternMetonyms.fit/2 within the do block.

0.6.1 - 2022-05-17
Added
	Now handling equality patterns.

0.6.0 - 2021-11-03
Added
	PatternMetonyms.view/2 can now use any function in guards, as long as they return true.
	Added PatternMetonyms.defv/2 to define named functions capable of the same things as PatternMetonyms.view/2.
	Added PatternMetonyms.fnv/1 do defined anonymous functions capable of the same things as PatternMetonyms.view/2.

0.5.0 - 2021-10-20
Changed
	Implicitly replace patterns defined using no args with patterns defined using an empty list of args.
	Improved the localisation of the pattern mentioned in the error message for patterns that can only be used inside PatternMetonyms.view/2.
	Breaking : require parentheses on raw views used with PatternMetonyms.view/2, and in views used with PatternMetonyms.pattern/1.
	Reworked PatternMetonyms.view/2 to handle nested pattern usage

0.4.0 - 2021-05-14
Changed
	Implementation details, using new dependency: Circe.
	Doc updated.

0.3.1 - 2021-03-11
Changed
	Doc updated.

Added
	Anonymous functions can now be used with PatternMetonyms.view/2.

0.3.0 - 2020-12-18
Changed
	Updated this changelog to try the format from keep-a-changelog.
	Reworked PatternMetonyms.view/2 to handle remote calls, updated the doc accordingly.

0.2.0 - 2020-12-18
Changed
	Docs updated.
	Reorganised the order of the patterns definitions, this allows to set the doc attribute for the version used by the programmer, and to hide the internal version used by PatternMetonyms.view/2.
	Added a vanilla definition for unidirectional pattern using a view, this permits setting a doc attribute usable by the programmer, improves discoverability, and signal with a better error message when the pattern is improperly used (from unknown definition to custom raise).
	Internal changes.

Added
	Guards can now be used with PatternMetonyms.view/2.

0.1.1 - 2020-12-04
Fix
	PatternMetonyms.view/2 updated to use a unique variable before matching, avoiding repeated function call when it is the value to be matched on.
	PatternMetonyms.pattern/1 : changed ast traversal from pre to post to avoid infinite loops during compilation (was present but forgotten in the changelog).

Changed
	Docs updated.

Added
	Changelog creation.

0.1.0 - 2020-12-03
Added
	Base implementation.

Changelog format inspired by keep-a-changelog.


  

    
PatternMetonyms 
    



      
Reusing the description from the paper:
Pattern synonyms allow to abstract over patterns used in pattern matching,
notably by allowing to use computation instead of being limited to concrete data.
Pattern metonyms are an implementations of this, but because of the limitation of the language,
it can not be considered the same, so metonyms was chosen as a synonym to synonyms.
Unlike in Haskell, few metonym definitions can be used with case/2,
other forms of patterns can only be used in combination with view/2.

        View with named functions



    


    
      
        fit(block)

      


        Replaces all occurences of =/2 in the do block by fit/2



    


    
      
        fit(pat, expr)

      


        =/2 equivalent to view, named form.



    


    
      
        fnv(clauses)

      


        View with anonymous functions



    


    
      
        matchv?(pat, expr)

      


        view/2's equivalent to match?/2



    


    
      
        pattern(metonym)

      


        Macro used to define pattern metonyms.



    


    
      
        view(expr, clauses)

      


        Macro substitute for case/2 capable of using pattern metonyms.

View with named functions
use PatternMetonyms

defv id((Function.identity() -> x)), do: x

Replaces all occurences of =/2 in the do block by fit/2
iex> import PatternMetonyms
...> fit do
...>   (Kernel.+(1) -> x) = 1
...>   (Kernel.-(1) -> y) = 41
...> end
...> x + y
42

iex> import PatternMetonyms
...> fit do
...>   (Kernel.+(1) -> x) = 1
...>   (Kernel.+(1) -> y) = x
...> end
...> x + y
5

iex> import PatternMetonyms
...> fit do
...>   x = 1
...>   y = x
...> end
...> x + y
2

iex> import PatternMetonyms
...> fit(do: (Kernel.+(1) -> x) = (Kernel.-(1) -> y) = 21)
...> x + y
42

=/2 equivalent to view, named form.
iex> import PatternMetonyms
...> pat = fn {:Just, x} -> x end
...> fit((pat.() -> n), {:Just, 3})
...> n + 1
4
Underscore prefixed variables are ignored, they are therefore unusable outside
fit/2's first argument.
Because of the lack of operator with the same associativity and precedence as =/2,
no operator form of fit/2 is planned. It would otherwise result in unintuitive behaviors.
Left nesting of fit/2 will result in a compilation error.

View with anonymous functions
import PatternMetonyms

id = fnv do (Function.identity() -> x) -> x end

view/2's equivalent to match?/2
iex> import PatternMetonyms
...> matchv?((Function.identity() -> 1), 1)
true

iex> import PatternMetonyms
...> matchv?((Function.identity() -> {1, _}), {1, 2})
true

iex> import PatternMetonyms
...> map = %{a: 1, b: 2}
...> matchv?((Map.take([:a, :b, :c]) -> %{a: _}), map)
true

iex> import PatternMetonyms
...> map = %{a: 1, b: 2}
...> matchv?((Map.take([:b, :c]) -> %{a: _}), map)
false

iex> import PatternMetonyms
...> a = 1
...> matchv?(^a, 1)
true

iex> import PatternMetonyms
...> a = 1
...> matchv?((Function.identity() -> ^a), 1)
true

iex> import PatternMetonyms
...> list = [a: 1, b: 2, a: 3]
...> Enum.filter(list, &matchv?((Function.identity() -> {:a, _}), &1))
[a: 1, a: 3]

iex> import PatternMetonyms
...> list = [a: 1, b: 2, a: 3]
...> Enum.filter(list, &matchv?((Function.identity() -> {:a, x}) when x < 2, &1))
[a: 1]

iex> import PatternMetonyms
...> matchv?(_x, 1)
...> binding()
[]

Macro used to define pattern metonyms.
There are three types of pattern, listed below with examples:
	Implicitly Bidirectional

The simplest type of pattern, because of its symmetry requirement, it can only
be defined using concrete data (or adapted macro). Therefore no computation is
allowed, and they are thus compatible with case and function heads.
They take the form:
pattern <name>(<[variables]>) = <pattern>
where pattern reuses the variables
iex> defmodule DoctestTPX do
...>   import PatternMetonyms
...>
...>   pattern ok(x)    = {:ok, x}
...>   pattern error(x) = {:error, x}
...>
...>   pattern cons(x, xs) = [x | xs]
...>
...>   def foo(x) do
...>     view x do
...>       ok(a)    -> a
...>       error(b) -> b
...>     end
...>   end
...>
...>   def bar(x) do
...>     case x do
...>       ok(a)    -> a
...>       error(b) -> b
...>     end
...>   end
...>
...>   def baz(ok(a)   ), do: a
...>   def baz(error(b)), do: b
...>
...>   def mk_ok(x), do: ok(x)
...>
...>   def blorg(xs) do
...>     view xs do
...>       cons(x, xs) -> cons(x, Enum.map(xs, fn x -> -x end))
...>     end
...>   end
...> end
iex> DoctestTPX.foo({:ok, :banana})
:banana
iex> DoctestTPX.foo({:error, :split})
:split
iex> DoctestTPX.bar({:ok, :peach})
:peach
iex> DoctestTPX.baz({:error, :melba})
:melba
iex> DoctestTPX.mk_ok(:melba)
{:ok, :melba}
iex> DoctestTPX.blorg([1, 2, 3])
[1, -2, -3]
	Unidirectional

This type of pattern is read only, it may be used as abstraction over pattern matching on concrete data type
that can not be reused to construct data, or as abstraction over views, as explained in view/2.
They take the form:
pattern <name>(<[variables]>) <- <pattern>
pattern <name>(<[variables]>) <- (<function> -> <pattern>)
where pattern reuses the variables.
(function -> pattern) is called a view
iex> defmodule DoctestTPY do
...>   import PatternMetonyms
...>
...>   pattern head(x) <- [x | _]
...>
...>   pattern rev_head(x) <- (reverse() -> head(x))
...>
...>   def reverse(xs), do: Enum.reverse(xs)
...>
...>   def foo(xs) do
...>     view xs do
...>       head(x) -> x
...>       []      -> []
...>     end
...>   end
...>
...>   def bar(x) do
...>     case x do
...>       head(a) -> a
...>       []      -> []
...>     end
...>   end
...>
...>   def baz(head(a)), do: a
...>
...>   def blorg(xs) do
...>     view xs do
...>       rev_head(x) -> x
...>     end
...>   end
...> end
iex> DoctestTPY.foo([1, 2, 3])
1
iex> DoctestTPY.bar([1, 2, 3])
1
iex> DoctestTPY.baz([1, 2, 3])
1
iex> DoctestTPY.blorg([1, 2, 3])
3
	Explicitly bidirectional

This type of pattern allows the same kind of abstraction as unidirectional one, but also permit defining
how to construct data from computation (if necessary).
They take the form:
pattern (<name>(<[variables]>) <- (<function> -> <pattern>)) when <name>(<[variables]>) = <builder>
where pattern and builder reuse the variables.
(function -> pattern) is called a view
iex> defmodule DoctestTPZ do
...>   import PatternMetonyms
...>
...>   pattern (snoc(x, xs) <- (unsnoc() -> {x, xs}))
...>     when snoc(x, xs) = Enum.reverse([x | Enum.reverse(xs)])
...>
...>   defp unsnoc([]), do: :error
...>   defp unsnoc(xs) do
...>     [x | rev_tail] = Enum.reverse(xs)
...>     {x, Enum.reverse(rev_tail)}
...>   end
...>
...>   def foo(xs) do
...>     view xs do
...>       snoc(x, _) -> x
...>       []      -> []
...>     end
...>   end
...>
...>   def bar(xs) do
...>     view xs do
...>       snoc(x, xs) -> snoc(-x, xs)
...>       []      -> []
...>     end
...>   end
...> end
iex> DoctestTPZ.foo([1, 2, 3])
3
iex> DoctestTPZ.bar([1, 2, 3])
[1, 2, -3]
Patterns using a view can not be used with case.
Remote function can be used within a view, but the __MODULE__ alias won't work
because the expansion is not done at the usage site. It is not yet determined
which behavior is desired.
iex> defmodule DoctestTPA do
...>   import PatternMetonyms
...>
...>   pattern rev_head(x) <- (Enum.reverse -> [x | _])
...>
...>   def blorg(xs) do
...>     view xs do
...>       rev_head(x) -> x
...>     end
...>   end
...> end
iex> DoctestTPA.blorg([1, 2, 3])
3
Unknown yet if anonymous functions can be supported.
Guards within a pattern definition is considered undefined behavior,
it may work, but it depends on the context.
Consider that if the behavior gets a specification, it would be the removal of
the possibility of using them. Patterns using a view pattern are the recommend
approach. For example:
pattern heart(n) <- (less_than_3 -> {:ok, n})
Patterns can be documented:
@doc """
heart matches when the number is heartfelt <3
"""
pattern heart(n) <- (less_than_3() -> {:ok, n})
You can then access the doc as usual: h heart, or h Module.heart.

Macro substitute for case/2 capable of using pattern metonyms.
Custom case able to use pattern metonyms defined with this module.
Largely unoptimized, try to avoid side effect in your pattern definitions as using them multiple time
in view will repeat them, but might not later on.
View pattern ((function -> pattern)) may be used raw in here.
View patterns are simply a pair of a function associated with a pattern
where the function will be applied to the data passed to view
and the result will be matched with the pattern.
iex> import PatternMetonyms
iex> view self() do
...>   (is_pid() -> true) -> :ok
...>   _ -> :ko
...> end
:ok
Guards can be used outside of the view pattern or the pattern metonym.
iex> import PatternMetonyms
iex> view -3 - :rand.uniform(2) do
...>   (abs() -> x) when x > 3 -> :ok
...>   (abs() -> x) when x < 3 -> :ko
...>   x -> x
...> end
:ok
Remote calls can be used directly within a view pattern.
iex> import PatternMetonyms
iex> view :banana do
...>   (Atom.to_string() -> "ba" <> _) -> :ok
...>   _ -> :ko
...> end
:ok
Anonymous functions can be used within a view pattern.
They can be either used as stored within a variable:
iex> import PatternMetonyms
iex> fun = &inspect(&1, pretty: &2)
iex> view :banana do
...>   (fun.(true) -> str) -> str
...> end
":banana"
Or defined directly using Kernel.SpecialForms.fn/1 only:
iex> import PatternMetonyms
iex> view 3 do
...>   (fn x -> x + 2 end -> n) -> n + 1
...> end
6

